

Beim Mikrocomputer KIM-1 kann während des Austestens von Programmen nur sehr umständlich die Wirkung des zuletzt ausgeführten Befehls auf CPU-interne Register und andere Speicherzellen verfolgt werden. Dieser Beitrag zeigt, wie durch eine einfache Zusatzschaltung und ein Interruptprogramm mit nur einem Tastendruck der Inhalt von mehreren Speicherzellen angezeigt werden kann.

Dr. Rainer Gerlich

Testhilfe für den KIM-1

So funktioniert's

Das Monitorprogramm des KIM-1-Systems zeigt immer nur die Adresse einer Speicherzelle mit deren Inhalt an. Möchte man während des schrittweisen Programmtestens im Single-Stop-Modus (SST-Modus) den Inhalt anderer Speicherzellen einblenden und das mühsame Eingeben neuer Adressen vermeiden, so besteht folgende Möglichkeit: Der Tester teilt durch ein geeignetes Signal mit, daß er entweder den Inhalt einer bestimmten Speicherzelle sehen oder in das Testprogramm an die Stelle des nächsten auszuführenden Befehls zurückverzweigen will. Solche Funktionen werden am besten durch einen Interrupt ausgelöst. Da das Monitor-Programm des KIM-1 das Interrupt-Disable-Bit im Statusregister setzt, kann dazu aber nur der NMI-Interrupt benutzt werden.

Im SST-Modus werden nur Befehlsfolgen, die im Bereich 1C00...1FFF stehen, ungehindert ausgeführt. In allen anderen Speicherbereichen wird nach der Ausführung eines Befehles ein NMI-Interrupt erzeugt, der eine Unterbrechung der Befehlsausführung und ein Verzweigen in das Monitor-Programm bewirkt. Damit das zur Testhilfe benötigte Interruptprogramm durchgehend abgearbeitet werden kann, ist daher ein kleiner Eingriff auf der KIM-Pla-

tine erforderlich. Außerdem muß ein Flag gesetzt werden, damit zwischen normalen „System“-Interrupts und den „Anzeigewechsel“-Interrupts unterschieden werden kann. Die dazu notwendigen Hardware-Änderungen zeigt *Bild 1*; das entsprechende Impulsdiagramm geht aus *Bild 2* hervor. *Bild 3* zeigt die Änderungen auf der KIM-Platine.

Zwei Programmbeispiele, mit denen durch nur einen Tastendruck eine Adresse oder mehrere Adressen nacheinander ohne große Mühe angezeigt werden können, werden ebenfalls beschrieben. Zum Einsprung in das Interruptprogramm müssen die Speicherzellen 17FA und 17FB folgendermaßen besetzt werden: 17FA mit 80 und 17FB mit 17.

Die Software

Aus der Vielzahl der Möglichkeiten sollen hier zwei Programme vorgestellt werden, die der einzelne Anwender leicht seinen Erfordernissen anpassen kann.

In beiden Programmen wird zunächst analysiert, ob der NMI-Interrupt vom Tester selbst oder von KIM ausgelöst wurde. Zur Unterscheidung wird Bit 7 von Port A (PA7) herangezogen. Hat PA7 nach der Verzweigung in das Interrupt-Programm den Zustand ‚L‘, so

wird ein normaler Systeminterrupt angenommen und in das Monitor-Interrupt-Programm verzweigt.

Im Programm (*Bild 4*) wird bei jedem Test-Interrupt die Variable CNT inkrementiert und in Abhängigkeit von ihrem Wert eine Adresse in den Pointer 00FA/00FB (POINTL, POINTH) für die Display-Anzeige durch indizierte Adressierung mit dem X-Register geladen. Die Adresse und der Inhalt der Speicherzelle erscheinen dann in der Anzeige. Auf diese Weise kann der Tester sich nacheinander in einer von ihm gewählten Reihenfolge den Inhalt verschiedener Zellen ansehen. Unabhängig vom Interrupt-Programm kann mit Hilfe des Monitor-Programmes die Display-Adresse mit Hilfe der „+“-Taste weiter inkrementiert werden. Ist z. B. die erste anzuzeigende Adresse 00F1, so können durch Betätigen der „+“-Taste nacheinander auch die Inhalte der Register und des Akkumulators sichtbar gemacht werden. Bei Auslösen des nächsten Interrupts wird die nächste Adresse in den Display-Pointer geladen, danach kann wieder bei Bedarf die Adresse inkrementiert werden.

Liegt der Wert von CNT (17E4) nicht innerhalb des geforderten Bereiches (im Beispiel zwischen 0 und 9), so wird CNT auf 0 gesetzt. Hat CNT den Wert 0, so wird der momentane Wert von





Bild 1. Zusatzschaltung zur Unterdrückung eines NMI-Interrupts bei der Adressierung von Block K5 im SST-Modus (oben) und zur Kontaktprellung und Erzeugung eines NMI-Interruptes bzw. eines Signales für PA7

POINTL, POINTH nach 17E5, 17E6 (ADRL, ADRH) gerettet. Wenn CNT = 9 ist, wird die ursprüngliche Adresse wieder nach (POINTL, POINTH) übertragen und damit ins Testprogramm zurückverzweigt. Falls erforderlich, kann der Anwender die hier vorgegebene obere Grenze 9 abändern. Dazu ist die neue obere Grenze in die Speicherzelle 1798 anstelle von 09 einzutragen.

Bei Programmbeispiel 1 ist von Nachteil, daß immer alle vorgegebenen Stationen durchlaufen werden müssen, auch wenn momentan nur ein bestimmter Speicherbereich interessiert. Das wird in Beispiel 2 (Bild 5) vermieden. Bei diesem Programm kann der Taster durch eine Schalterstellung unter einer von mehreren Möglichkeiten auswählen. Beim ersten Interrupt wird die Basisadresse des gewünschten Speicherbereiches angezeigt und beim nächsten Interrupt erfolgt ein Sprung zurück in das zu testende Programm.

Da durch die vorgegebene Schalterstellung eine Prüfung auf zulässige Werte entfallen kann und somit mehr Speicherplatz zur Aufnahme des Interrupt-Programmes zur Verfügung steht, konnte bei diesem Programmbeispiel

Bild 3. Anschlußpunkte und Lage der Unterbrechungsstelle auf der KIM-1-Platine



KIM-1-Platine





Bild 2. Zeitliches Verhalten der Schaltung in Bild 1



auch noch die Ausgabe über TTY realisiert werden.

Durch einen externen Schalter werden die Bits 0, 1, 2 des Ports A (PA0, PA1, PA2) modifiziert, so daß unter insgesamt 8 Basisadressen ausgewählt werden kann. Da die Arbeitsvariablen CNT, BASISL, BASISH aus Platzgründen nicht mehr im selben Speicherbereich wie das Programm untergebracht werden konnten, können diese Variablen im Unterschied zu Programmbeispiel 1 nicht mit dem Laden des Programmes gleichzeitig vorbesetzt werden. Bevor das Programm zum ersten Mal nach „Power on“ benutzt wird, ist daher eine Definition dieser Variablen notwendig. Dies gilt besonders für die Variable CNT, die vor dem ersten Programmstart unbedingt den Wert Null zugewiesen bekommen muß.

Verläuft der TTY-Test positiv, so wird der Inhalt der ausgewählten Speicherzelle ausgegeben und anschließend noch während des gleichen Interrupts die Verzweigungsadresse wieder geladen. Daher ist bei TTY-Anschluß ein weiterer Interrupt zum Verzweigen ins Testprogramm nicht erforderlich. Zur TTY-Ausgabe werden die Systemprogramme PRTPTNT, PRTBYT, OUTSP, CRLF des KIM-1 verwendet [1].

Möchte der Benutzer mehr oder weniger Möglichkeiten zur Auswahl haben, so müssen dazu die Speicherzellen 1784 (Definition des Data Direction Registers für Port A) und 1797 (Maske für eingelesenes Byte) abgeändert werden. Die Adressen der Speicherzellen, deren Inhalt angezeigt werden soll, sind für

Programm 1 von 17D2 bis 17E3, für Programm 2 von 00DC bis 00EB einzutragen.

Erforderliche Hardware-Änderungen

Beim KIM-1 wird im SST-Modus mit Hilfe des SYNC-Signals, das das Laden des nächsten Operationscodes ankündigt, ein NMI-Interrupt erzeugt [2]. Die fortlaufende Bearbeitung des Operationscodes wird dann nach Ausführung eines Befehls immer unterbrochen. Ausgenommen davon sind Programme im Bereich 1C00...1FFF (Block K7). Jedemal, wenn eine Zelle in K7 adressiert wird, nimmt ein Eingang des NAND-Gatters U26, B den Low-Zustand an, so daß dann ein Interrupt verhindert wird. Um die ungehinderte Ar-

beit des Interrupt-Programmes für die Testunterstützung im SST-Modus zu ermöglichen, muß bei der Adressierung des Speicherbereiches, in den dieses Interrupt-Programm geladen wird, ebenfalls der SYNC-NMI-Interrupt unterdrückt werden.

Als Speicherbereich für das Interrupt-Programm bietet sich das vom KIM-1 nicht genutzte RAM auf den beiden Chips 6530 (U2 und U3 auf der KIM-Platine) an, d. h. der Speicherbereich 1780 bis 17E6. Dieses RAM ist der einzige (freie) Speicher im Block K5, so daß durch eine Sonderstellung dieses Blockes trotzdem andere Programme im übrigen Speicher weiter schrittweise ausgetestet werden können. Eine direkte „Wired-OR“-Verbindung zwischen K5 und K7 ist leider nicht möglich, da beide Signale entkoppelt bleiben müssen. Daher sind zwei NAND-Gatter zusätzlich erforderlich. Durch einen Taster wird nach Kontaktentprellung ein Ausgangsimpuls für PA7 und ein Interrupt-Impuls erzeugt.

```

1780 48          PHA
1781 A9 7F      LDA =7F
1783 2D 01 17  AND 1701
1786 8D 01 17  STA 1701
1789 AD 00 17  LDA 1700
178C 30 04      BMI 1792
178E 68          PLA
178F 4C 00 1C  JMP 1C00
1792 8A          TAX
1793 48          PHA
1794 AD E4 17  LDA 17E4
1797 C9 09      CMP =09
1799 D0 0D      BNE 17A8
179B AD E5 17  LDA 17E5
179E 85 FA      STA FA
17A0 AD E6 17  LDA 17E6
17A3 85 FB      STA FB
17A5 4C C9 17  JMP 17C9
17A8 90 05      BCC 17AF
17AA A9 00      LDA =00
17AC 8D E4 17  STA 17E4
17AF C9 00      CMP =00
17B1 90 F7      BCC 17AA
17B3 0A          ASL A
17B4 AA          TAX
17B5 D0 0A      BNE 17C1
17B7 A5 FA      LDA FA
17B9 8D E5 17  STA 17E5
17BC A5 FB      LDA FB
17BE 8D E6 17  STA 17E6
17C1 BD D2 17  LDA 17D2, X
17C4 85 FA      STA FA
17C6 BD D3 17  LDA 17D3, X
17C9 85 FB      STA FB
17CB EE E4 17  INC 17E4
17CE 68          PLA
17CF AA          TAX
17D0 68          PLA
17D1 40          RTI

```

Bild 4.
Programmbeispiel 1 mit Display-Ansteuerung

Beim Schließen des Schalters S wird am Ausgang von Gatter 2D ein Normimpuls erzeugt, dessen Länge vom Ladezustand von Kondensator C1 abhängt. Da C1 sich beim Schließen von S relativ schnell über R2 entlädt, aber nur langsam wieder (z. B. beim Prellen von S) durch R1 aufgeladen werden kann, ändert sich der Ausgangszustand von Gatter 2D bis auf die Einschwingphase nicht, da ein Eingang auf „L“ liegt, bis Schalter S wieder geöffnet wird. Nur während des kurzen Entladevorgangs beim Schließen von S kann sich das Kontaktprellen auf den Ausgang von Gatter 2D auswirken. Kondensator C2 kann jedoch erst dann voll aufgeladen werden, wenn der Schalter S eine längere Zeit nicht geöffnet ist. Daher wird am Ausgang von Gatter 2A bereits ein relativ sauberer Impuls erzeugt. Kondensator C3 integriert die verbliebenen, geringen restlichen Störungen weg. Nach Invertierung gelangt der Impuls zu PA7 (Application Connector, Pin 8) und nach Differenzierung an den NMI-Eingang (Expansion Connector, Pin 6). Das differenzierte Signal erzeugt einen NMI-Interrupt, das PA7-Signal steuert die Verzweigung innerhalb des Interrupt-Programms in den Teil, der den Anzeigewechsel bewirkt.

Die Länge des NMI-Impulses beträgt ca. 35 µs, der PA7-Impuls ist immer länger als 3 ms. Die maximale Wiederholungsrate für einen Testhilfe-Interrupt wird durch das Zeitglied R1/C1 bestimmt; sie liegt max. bei etwa 3 Hz. Wird der Schalter S öfter pro Zeiteinheit geschlossen, so wird kein Ausgangsimpuls erzeugt.

Beim Anschließen des KIM muß die Verbindung von U26, Pin 4 zu U4, Pin 9 unterbrochen werden. Am besten geschieht dies durch Auftrennen der Leiterbahn, die zwischen Pin 7 und 8 unter U26 hervortritt. Da von U4, Pin 9 noch eine Verbindung zu U2, Pin 4 besteht, sollte die Leiterbahn nicht in der Nähe von U4 unterbrochen werden. GND und +5 V können von Pin 7 bzw. Pin 14 von U26 (7438) abgenommen werden. Die Zusatzschaltung kann leicht auf einer kleinen Platine über der KIM-Platine angebracht werden.

Ausblick

Es soll nicht unerwähnt bleiben, daß auf diese Weise nicht nur Interrupt-Programme für die Speicherinhaltsanzeige realisiert werden können, sondern auch andere Testunterstützungen, z. B. die (Neu-)Definition von Variablen vor einem Testschritt durchgeführt werden können. Dazu aber noch ein Hinweis: Beim Behandeln eines Interrupts müssen Akkumulator und Register, falls sie im Interrupt-Programm benutzt werden, auf den Stack gerettet und vor dem Rücksprung wieder vom Stack geholt werden.

Literatur

- [1] KIM- und SYM-Monitor-Unterprogramme. FUNKSCHAU 1979, Heft 11, Seite 652
- [2] KIM-1 Users' Manual. MOS Technology Inc., 1976

Stichworte zum Inhalt

Single-Step-Modus, SST, KIM-1, 6502, Entprellung, Interrupt, Registeranzeige.

```

1780 48          PHA
1781 8A          TXA
1782 48          PHA
1783 A9 78      LDA =78
1785 2D 01 17  AND 1701
1788 8D 01 17  STA 1701
178B AD 00 17  LDA 1700
178E 30 06      BMI 1796
1790 68          PLA
1791 AA          TAX
1792 68          PLA
1793 4C 00 1C  JMP 1C00
1796 29 07      AND =07
1798 0A          ASL A
1799 AA          TAX
179A 98          TYA
179B 48          PHA
179C 24 EC      BIT EC
179E 30 37      BMI 17D7
17A0 C6 EC      DEC EC
17A2 A5 FA      LDA FA
17A4 85 ED      STA ED
17A6 A5 FB      LDA FB
17A8 85 EE      STA EE
17AA B5 DC      LDA DC, X
17AC 85 FA      STA FA
17AE B5 DD      LDA DD, X
17B0 85 FB      STA FB
17B2 A0 00      LDY =00
17B4 8C 41 17  STY 1741
17B7 A9 3F      LDA =3F
17B9 8D 43 17  STA 1743
17BC 8C 42 17  STY 1742
17BF A9 01      LDA =01
17C1 2C 40 17  BIT 1740
17C4 D0 1B      BNE 17E1
17C6 20 2F 1E  JSR 1E2F
17C9 20 1E 1E  JSR 1E1E
17CC 20 9E 1E  JSR 1E9E
17CF B1 FA      LDA (FA), Y
17D1 20 3B 1E  JSR 1E3B
17D4 20 9E 1E  JSR 1E9E
17D7 E6 EC      INC EC
17D9 A5 ED      LDA ED
17DB 85 FA      STA FA
17DD A5 EE      LDA EE
17DF 85 FB      STA FB
17E1 68          PLA
17E2 A8          TAY
17E3 68          PLA
17E4 AA          TAX
17E5 68          PLA
17E6 40          RTI

```

Bild 5. Programmbeispiel 2 mit TTY-Ausgabe